Investment Adjustment Costs

ECON 724: Macroeconomic Theory II

C. Aquino

J.D. Rios

05 May 2017

1 Introduction

Real Business Cycle (RBC) models are critical to the study of macroeconomics. In principle, basic RBC models are equivalent to neoclassical growth models with stochastic technological shocks. The emphasis on the stochastic component of the model is derived from the fact that RBC models are generally used to examine output dynamics in the presence of exogenous shocks. The hope is that RBC models will allow us to simulate the stylized output dynamics of the economic variables present in the data. In particular, a desirable RBC model will have an internal propagation mechanism that accurately produces persistence, comovement, and volatility in aggregate economic variables and a "hump-shaped" impulse response function after being exogenously shocked. This attempt to match multi-dimensional facts follows directly from King and Rebelo (2000) who focus on the size of technological shocks as they affect RBCs and Cogley and Nason (1995) who observe inconsistencies between the RBC literature and stylized facts.

The basic problem of neoclassical growth models and RBC models is to maximize the expected lifetime utility of an infinitely-lived representative household subject to a set of constraints. By examining different specifications of the agent's utility function or the set of economic constraints, a rich set of models can be considered. Among these candidate models, designing an internal propagation mechanism that solves pieces of this multi-dimensional problem is possible. This paper compares the benchmark RBC model with one such specification, namely, the presence of investment adjustment costs (IACs). The IAC framework is a specification of the RBC model that introduces a penalty for rapid changes in the level of capital investment between neighbouring periods. While the IAC framework produces persistence, comovement, and volatility that tends to agree with the general trends present in the data, the difference in performance in these dimensions when compared to the benchmark model is not significant. Instead, the main result of introducing IACs is the ability to generate weakly hump-shaped impulse response functions.

2 Model

2.1 Benchmark RBC Model

The benchmark RBC model is, generally speaking, not robust to the stylized economic facts. By naively specifying only a traditional aggregate resource constraint, a frictionless law of motion for capital accumulation, and an aggregate production function depending only on exogenous shocks and the current labour and capital stock, RBC models are not able to properly mimic real-world data. Some comovement, persistence, and variability is achievable. Variants of the RBC model include modelling habit formation in the utility function and organizational capital in a learning-by-doing framework. Both of these variants introduce frictions designed to encourage greater comovement of aggregate macroeconomic variables and cyclicality that is accurately reflected in the data. The benchmark RBC model is given by the following set of equations:

$$\max E_0 \sum_{t=0}^{\infty} \beta^t [\ln C_t + \psi \ln(1 - N_t)] \qquad \text{subject to} \tag{1}$$

$$Y_t = C_t + I_t \tag{2}$$

$$Y_t = A_t F(K_t, N_t) = A_t K_t^{\alpha} N_t^{1-\alpha} \tag{3}$$

$$K_{t+1} = I_t + (1 - \delta) K_t \tag{4}$$

$$1 = N_t + L_t \tag{5}$$

$$\ln A_{t+1} = \rho \ln A_t + \epsilon_{t+1} \mid \epsilon_{t+1} \overset{iid}{\sim} N(0, \sigma_\epsilon^2) \qquad \text{where} \tag{6}$$

$$\psi > 0,\ 0 < \beta < 1,\ 0 < \delta < 1,\ 0 < \rho < 1,\ K_0 > 0,\ A_0 > 0$$

Equation (1) is the utility function of our infinitely lived individual where ψ captures the relative value of leisure and consumption inutility terms, and β reflects the discount rate of future utility. Equation (2) is the aggregate resource constraint. Equation (3) is the production function that takes a Cobb-Douglas form with capital (K), labour (L) and a variable that captures total factor productivity (A). The law of motion for capital accumulation is given by equation (4) where next period's capital is equal to capital investment in the current period (I) and the undepreciated capital that survives into the next period. The depreciation rate is given by δ. The central planner's choice variables include current period consumption (C_t), investment (I_t), output (Y_t), and labour supplied (N_t) as well as the next periods capital stock (K_{t+1}) . Equation (5) describes the representative household's time constraint, and equation (6) is the law of motion for TFP shocks.

Solving the model involves pooling equation (2), equation (3), and equation (4) into one single constraint and optimizing by introducing a Lagrangian multiplier. The economics of the model will be described in detail in the model comparison section. Using Dynare, the endogenous variables in the benchmark model can be solved in terms of moments, autocorrelation coefficients, contemporaneous correlations with output, and steady-state values. These can be compared to the volatility, persistence, and comovement present in the data. The first step is to define all endogenous variables and fully parameterize the model. The parameterized values are $\alpha = 0.35,\ \beta = 0.97,\ \delta = 0.06,\ \rho = 0.95,\ \sigma_\epsilon = 0.01,\ \psi = 1$. This calibration follows Torres (2015) and is consistent with the DSGE literature. We solve the model in Dynare by inputting the following six equations that characterize the solution of the model (see the appendix for a detailed breakdown of the first-order conditions):

```
% Standard RBC Model

% Equation 1: Euler Equation for K
1/exp(c)          = beta*(1/exp(c(+1)))*(alpha*exp(A(+1))*exp(k)^(alpha-1)*exp(n(+1))^(1-alpha) + (1-delta));

% Equation 2: Euler Equation for N
psi/(1-exp(n))    = (1/exp(c))  * ((1-alpha)*exp(A)*exp(k)^(alpha)*exp(n)^(-alpha));

% Equation 3: Production Function
exp(y)            = exp(A)*(exp(k(-1))^(alpha))*exp(n)^(1-alpha);

% Equation 4: Resource Constraint
exp(y)            = exp(c) + exp(k)  - (1-delta)*exp(k(-1));

% Equation 5: Law of Motion for TFP Shocks
A                 = rho*A(-1) + e;

% Equation 6: Law of Motion for Capital Accumulation
exp(i)            = exp(k)  - (1-delta)*exp(k(-1));
```

A complete version of the Matlab code needed to solve the RBC model is included in the appendix.

2.2 Investment Adjustment Cost Model

The investment adjustment cost model introduces frictions to altering the level of investment between neighbouring periods. The purpose of introducing frictions, according to Christiano, Eichenbaum, and Evans (2005) is to enable the benchmark model to produce greater comovement in the presence of stochastic technological shocks. Investment adjustment costs are one such friction. The following investment adjustment cost model follows the same structure as the RBC benchmark model with the exception of the law of motion for the accumulation of capital stock. The capital accumulation equation changes such that next period's capital stock is equal to the current period's undepreciated capital stock and a function of investment relating current and past investment levels. This is intentionally done to isolate changes that introducing an investment adjustment cost has on the model.

$$\max E_0 \sum_{t=0}^{\infty} \beta^t [\ln C_t + \psi \ln(1 - N_t)] \qquad \text{subject to} \tag{7}$$

$$Y_t = C_t + I_t \tag{8}$$

$$Y_t = A_t F(K_t, N_t) = A_t K_t^{\alpha} N_t^{1-\alpha} \tag{9}$$

$$K_{t+1} = \big[1 - \Phi(\frac{I_t}{I_{t-1}})\big] I_t + (1 - \delta) K_t \tag{10}$$

$$1 = N_t + L_t \tag{11}$$

$$\ln A_{t+1} = \rho \ln A_t + \epsilon_{t+1} \mid \epsilon_{t+1} \overset{iid}{\sim} N(0, \sigma_{\epsilon}^2) \qquad \text{where} \tag{12}$$

$$\Phi(1) = 0,\ \psi > 0,\ 0 < \beta < 1,\ 0 < \delta < 1,\ 0 < \rho < 1,\ K_0 > 0,\ A_0 > 0$$

The aggregate resource constraint, production function, law of motion for TFP shocks, the individual's time constraint remain the same as the benchmark RBC model.The law of motion for capital in equation (10) now includess a function that penalizes dramatic changes between I_t and I_{t-1}. This function has the property that $\Phi(1) = 0$ when $I_t = I_{t-1}$ and the law of motion for capital reverts back to the benchmark RBC model. To get an analytical solution to the model, the penalized change in investment function must follow a specific functional form. Following Uribe and Schmitt-Grohé (2005), one often used functional form is a quadratic specification:

$$\Phi(\frac{I_t}{I_{t-1}}) \equiv \frac{\phi}{2} \left[\frac{I_t}{I_{t-1}} - 1\right]^2$$

Where $\phi > 0$ represents relative size of the investment adjustment penalty and examining different values of ϕ will be particularly useful for sensitivity checks. Using the Lagrangian method, this model will have two constraints. The introduction of a law of motion for capital accumulation that is not additively separable in investment eliminates the ability to recursively substitute investment out of the problem as a choice variable. In the investment adjustment cost framework, investment is an endogenously determined choice variable that will have its own first-order condition and associated Euler equation. For the sake of interpretation of equation (6) in the Dynare code the first-order condition for current period captial investment is given by:

$$\lambda_t = \omega_t \Bigg[1 - \frac{\phi}{2}\big(\frac{I_t}{I_{t-1}} - 1\big)^2 - \phi\frac{I_t}{I_{t-1}}\big[\frac{I_t}{I_{t-1}} - 1\big]\Bigg] + \beta\omega_{t+1}\Bigg[\phi\bigg(\frac{I_{t+1}}{I_t}\bigg)^2\bigg(\frac{I_{t+1}}{I_t} - 1\bigg)\Bigg]$$

The endogenous variables of the model are $Y_t,\ C_t,\ I_t,\ K_t,\ N_t,\ A_t$ with two Lagrange mutlipliers $\lambda_t,\ \omega_t$. We can interpret λ_t as the societal value of an additional unit of consumption and ω_t as the societal value of an additional unit of capital. The parameterized values are specified to be the same as in the benchmark model, namely, $\alpha = 0.35,\ \beta = 0.97,\ \delta = 0.06,\ \rho = 0.95,\ \sigma_\epsilon = 0.01,\ \psi = 1$. Initially, a value of $\phi = 4.5$ is chosen but our sensitivity analysis will consider a variety of specifications. The economics of the model will be described in detail in the model comparison section. The model solution is described by the following set of eight equations:

```
% IAC Model;

% Equation 1: FOC for C
1/exp(c) = exp(lambda);

% Equation 2: FOC for N
psi/(1-exp(n)) = exp(lambda)*(1-alpha)*(exp(y)/exp(n));

% Equation 3: Law of Motion for TFP Shocks
A = rho * A(-1) + e;

% Equation 4: Production Function
exp(y)= exp(A)*(exp(k(-1))^alpha)*exp(n)^(1-alpha);

% Equation 5: Resource Constraint
exp(y) = exp(c) + exp(i);

% Equation 6: Law of Motion for Capital Accumulation (Penalized)
```

```
exp(k) = exp(i) - (phi/2)*exp(i)*((exp(i)/exp(i(-1))) -1)^2 + (1-delta)*exp(k(-1)) ;

% Equation 7: FOC for K
exp(omega) = beta*exp(lambda(+1))*(alpha*(exp(y(+1))/exp(k)))+beta*exp(omega(+1))*(1-delta);

% Equation 8: FOC for I
exp(lambda) = exp(omega)-exp(omega)*(phi/2)*((exp(i)/exp(i(-1))-1)^2)-exp(omega)*phi*( (exp(
    i)/exp(i(-1))) -1 )* (exp(i)/exp(i(-1))) + beta*exp(omega(+1))*phi*((exp(i(+1))/exp(i))
    -1)*((exp(i(+1))/exp(i))^2);
end;
```

A complete version of the Matlab code needed to solve the IAC model is included in the appendix.

3 Results

3.1 Benchmark RBC Model

Reported below are the key results from the specification of the benchmark RBC model and the associated impulse reponse functions. The impulse response functions show the time-path of endogenous variables with respect to percentage deviations from their steady-state values after the model experiences an exogenous 1% shock to total factor productivity. The interpretation of the results will be described in the model comparison section.

Table 1: Business Cycle Statistics for Benchmark RBC Model

Variable	Steady-State Value	Standard Deviation	Rel. Standard Deviation	Contemporaneous Correlation with Output	First Order Auto-correlation
Y	-0.055	0.050	1.000	1.000	0.969
C	-0.317	0.044	0.881	0.970	0.991
I	-1.522	0.083	1.648	0.901	0.905
K	1.293	0.052	1.036	0.919	0.997
N	-0.781	0.007	0.141	0.562	0.848
A	0.000	0.032	0.634	0.989	0.950

Figure 1: Impulse Response Functions for Benchmark RBC Model



3.2 Investment Adjustment Cost Model

Reported below are the key results from the specification of the IAC model with $\phi = 4.5$ and the associated impulse response functions.

Table 2: Business Cycle Statistics for IAC Model ($\phi = 4.5$)

Variable	Steady-State Value	Standard Deviation	Rel. Standard Deviation	Contemporaneous Correlation with Output	First Order Auto-correlation
Y	-0.055	0.045	1.000	1.000	0.977
C	-0.318	0.042	0.931	0.991	0.966
I	-1.521	0.060	1.319	0.949	0.990
K	1.293	0.043	0.960	0.866	0.998
N	-0.781	0.004	0.080	0.518	0.948
A	0.000	0.032	0.710	0.972	0.950
λ	0.318	0.042	0.931	-0.991	0.966
ω	0.318	0.038	0.838	-0.955	0.995

Figure 2: Impulse Response Functions for IAC Model



3.3 Model Comparison and Economic Interpretation

When comparing the summary statistics in Table 1 and Table 2, it is clear that both RBC and the IAC models produce the same steady state values given the same parameters. This is a result of the fact that in steady state investment does not vary between periods and thus there is no investment adjustment penalty. The primary difference between the two models is their behaviour outside of steady state. The interpretation of each key endogenous variable in both models is described below:

Investment (I): The investment impulse response function in the benchmark model is a decreasing convex function that rises dramatically in the period following the productivity shock and decays slowly over time. Immediately following the shock, the central planner observes an increased marginal product of capital. Because productivity shocks follow an autoregressive process, this translates to above steady-state values for the marginal product of capital in future periods as well and creates an incentive for the central planner to increase the level of capital investment. In subsequent periods, the marginal product of capital returns to its steady-state value which results in an ever-decreasing level of capital investment. This autoregressive dynamic results in an impulse response function for investment following the same shape as the impulse response function of the productivity shock.

In contrast, the investment impulse response function in the IAC framework has a hump-shape with a smaller initial increase in the period following the productivity shock. The economic dynamics can be easily explained by equation (38). In words, there are three economic forces driving investment decisions. First, the productivity shock increases the marginal product of capital and incentivizes increases in the level of capital investment. Second, the quadratic investment adjustment cost function penalizes large changes in the level of capital investment away from its steady-state value and disincentivizes increases in the level of capital investment. Third, the investment adjustment cost function has an intertemporal component such that investment in the current period reduces the adjustment cost of maintaining a high level of investment in subsequent periods. Overall, these competing incentives result in an increased level of capital investment that is smaller than in the frictionless RBC model. In the subsequent period, the central planner observes a smaller marginal product of capital closer to steady state but chooses to increase investment. While this may seem contradictory, the reasoning for this is that last period's jump in investment makes increasing investment this period significantly less expensive. We can interpret the increase in investment until the peak of the curve as the central planner spreading out a rise in investment over multiple periods due to the quadratic investment adjustment cost term. Eventually the falling marginal product of capital does not warrant increasing investment, and the central planner begins to decrease investment similarly to the RBC benchmark. An important note as that the central planner decreases investment at a smaller rate than the RBC benchmark model again because of the cost associated with large changes in investment. These result are demonstrated in the relative standard deviation to output of investment in Table 1 and Table 2. Due to the investment adjustment cost the IAC has a smaller deviation to output than the RBC benchmark model.

Capital (K): The difference between the IAC and RBC capital impulse response functions directly follows from how each model translates investment into future capital. After a shock, the RBC benchmark model has a higher value of capital for two reasons. First, the benchmark model experiences a higher level of capital investment than the IAC model in the first period. Second, the condition that adjustment penalty $\phi > 0$ implies that the IAC model does not directly translate investment into capital at a one-to-one ratio. The reason that capital continues to accumulate even as investment is falling is explained by the fact that there is not full capital depreciation and some of the capital invested survives into subsequent periods. The IAC model follows the same shape of the RBC benchmark model but is lower due to the fact that there is less investment in IAC. The IAC model follows the same shape of the RBC benchmark model but is lower due to the investment adjustment penalty and the fact that there is less investment in the IAC model.

Output (Y): From the output impulse response function, it is clear that a productivity shock creates a smaller increase in output in the IAC model in comparison to the RBC benchmark model. This gap in production is a direct consequence of the friction introduced by the investment adjustment cost function. The central planner is unable to increase investment to reflect the increased marginal product of future capital to the same extent as in the benchmark RBC model. The reduced investment in capital accounts for the inital gap between the output of the IAC model and RBC benchmark model.

Consumption (C): The impulse response for consumption in the RBC benchmark model is a hump-shaped function that is driven by several economic forces in the model. Primarily the technology shock results in increased output which results in an initial increase in consumption. The log nature of the utility specification incentivizes consumption smoothing. While the increased marginal product of future capital incentivizes the central planner to forgo consumption directly following the productivity shock and instead invest in capital for increased consumption in the future. In the IAC model we observe higher initial consumption due to the fact that output is higher. The investment adjustment cost prohibits high levels of investment and instead results in consuming a larger portion of the initial output. In subsequent periods the increasing investment, as well as the reduced output, result in a lower consumption level than the RBC benchmark model. The investment adjustment cost stops the central planner from spreading consumption to future periods and the lower levels of production limits the consumption in comparison to the RBC benchmark model.

Labour Supply (N): In the RBC benchmark model, we observe an increase in the marginal product of labour due to the productivity shock. This incentivizes an increase in labour supplied. Counterbalancing this is a decrease in the societal value of an additional unit of consumption, due to increased consumption, pushing the labour supply downwards. These two effects result in a small increase in the amount of labour supplied. In subsequent periods the reduced marginal product of labour as well as increased consumption incentivizes labour supply levels closer to steady state. The IAC model has an initial decrease in the labour supplied. This is due to the fact that the decrease in the societal value of an additional unit of consumption is much larger due to a larger increase in consumption. As consumption falls and the marginal product of labour is above steady state, the central planner chooses to increase the labour supply. This creates a hump-shaped impulse response function with an initial decrease in the labour supply.

Lambda/Omega (λ/ω): We can see that the IAC model experiences a decrease in the ratio of the societal value of an additional unit of consumption divided by the societal value of an additional unit of capital. While this ratio remains constant in the RBC benchmark model. This is due to the fact that the investment adjustment cost prohibits a large change in investment to reflect the increased marginal product of future capital and the intial increased output is instead consumed. This decreases λ and the increased marginal product of capital increases ω.

3.4 Sensitivity Checks

Next, we consider the results generated by our model under different parameterizations of the adjustment penalty ϕ. First, we consider a result estimated by Smets and Wouters (2003) of $\phi = 5.9$. Then, for the purposes of understanding the penalty's influence on the impulse response functions, we consider two extreme values of ϕ, namely, $\phi = 50$ and $\phi = 500$. The RBC model is included as well as the benchmark case is a good approximation for $\phi = 0$. Reported below are the impulse response functions under different specifications of ϕ.

Figure 3: Impulse Response Functions for IAC Model Under Different Penalty Parameterizations



The first clear result is that as ϕ increases, the incentive to invest falls. The motivation for this is that as ϕ increases even small changes in the level of investment are penalized. Resultantly, even though the marginal productivity of capital is higher, the cost of capturing the higher productivity becomes higher. Thus, the level of investment as a percentage deviation from its steady-state value falls. It follows that the capital stock will fall and lead to output and eventually consumption falling as well. For extreme values of the penalty parameter, such as $\phi = 500$, the incentive to invest above steady-state falls almost to zero and capital, output and consumption adjust accordingly.

An interesting view of this phenomenon can be seen in the ratio of the marginal utilities of consumption and investment, λ/ω. For low values of ϕ the number of periods required for the marginal utilities to be approximately equal happens much more quickly than under large values of ϕ. The motivation for this is clear when the ratio of λ/ω is less than one. In this case, the investment adjustment cost penalty is significantly inhibiting the central planner from choosing the level of investment that they would if there were no investment adjustment costs. This means that relaxing the Lagrange multiplier on investment would result in a larger increase in investment than relaxing the Lagrange multiplier on consumption.

3.5 Paper Comparison

We compare our results of the differences between the benchmark RBC model and the IAC model to the results found in "Lumpiness, capital adjustment costs and investment dynamics" by Fiori (2012). Fiori undertakes a similar exercise as our simulations in comparing whether the introduction of lumpiness in capital accumulation or the introduction of a convex capital adjustment cost to a neoclassical two-sector model helps replicate patterns found in the macroeconomic time series data. The author describes lumpiness as a phenomenon seen in the empirical literature. More specifically, prior research by Doms and Dunne (1998) has shown that lumpiness can imply that as much as half of all plant-level capital investment in a 17-period horizon can take place in one period. Hence, this is a motivation for the existing research on RBC and IAC models from an empirical point of view.

Resultantly, lumpiness is another form of an investment friction similar to an investment adjustment cost. This permits comparison between these types of models. Lumpiness can be characterized as an investment friction that prohibits firms from over-investing for reasons other than an explicitly stated penalized ϕ parameter. It is clear, however, that this type of friction will have a similar effect to the frictions faced in our simple IAC framework. This allows us to compare the impulse response functions in a two-sector lumpy model to our simple IAC model in response to an exogenous shock. The notion of a hump-shaped impulse response function is consistent with the lumpy plant-level investment data.

The second model tested by Fiori is the convex capital adjustment costs model. This model more closely resembles the investment adjustment cost model. Instead of penalizing deviations from the previous period's level of capital investment as in the IAC model, Fiori uses a capital adjustment function that penalizes the difference between investment in the current period and the capital stock. Fiori's model has a ϕ parameter that reflects the size of adjustment costs, and like our model, adjustment costs are zero in the steady-state.

The simulated impulse response functions in the author's model closely resemble the results we have demonstrated in our simulation. Specifically, in a frictionless set-up all endogenous variables deviate above their steady state values. This is consistent with the fact that a positive TFP shock increases both the marginal productivity of capital and the marginal productivity of labour above their steady-state levels. Moreover, the impulse responses in the lumpy framework follow weakly hump-shaped curves that reflect the fact that investment is not rapid or gradual from period-to-period but is instead lumpy and sporadic. This notion holds in both simulations as well as the data and demonstrates the idea that investment is, generally speaking, fixed in the short run. The majority of investment decisions take place sporadically and idiosyncratically.

The reported business cycle volatilities also loosely reflect the results of our simulation. The IAC model and the lumpy two-sector model presented by the authors both understate the relative volatility of investment compared to the data and overstate the contemporaneous correlation with consumption and output. This comparison between Fiori's lumpy two-sector investment model and our IAC model show that there is some unification in RBC theory surrounding the use of frictions to induce internal propagation that mimics the stylized economic facts. Moreover, this comparison of models proves that our results are consistent with a brief survey of the literature.

4 Conclusion

This paper has considered the effects of investment adjustment costs on real business cycle models. Any dynamic stochastic general equilibrium model without any cost frictions is capable of replicating some of the stylized economic facts present in the data, but as we have shown, there are many avenues for researchers to improve upon the benchmark model. RBCs do in fact explain some of the multi-dimensional problems in moment-matching. Namely, RBCs accurately produce some comovement, persistence and volatility that is shown in the data. However, the benchmark RBC does not generate desirable hump-shaped impulse response functions in response to being exogenously shocked. Alternatively, the investment adjustment cost model has an internal propagation mechanism that produces a weakly hump-shaped impulse response function.

Much of what drives the shape of the impulse response function in an IAC framework is the penalty parameter ϕ. Adjusting the penalty parameter has little effect on affecting the correlation of aggregate variables with output, the persistence in the auto-correlation function, or the volatility. The ordering, for instance that the relative volatility of investment is higher than output, remains intact. Unsurprisingly, specifying an arbitrarily low penalty parameter results in the model reverting back to the benchmark case. Likewise, even different and more realistic parameterizations of the penalty yields an impulse response function that does not appear to change very much. Perhaps the biggest result of introducing investment adjustment costs can be seen graphically in Figure 2 in the stark difference in impulse response functions with and without parameterization.

There are many economic forces driving these results. In response to an exogenous shock, all endogenous variables tend to deviate to levels above their steady-state values. The transitional dynamics of the system directly follow from the new set of incentives faced by the central planner. For example, a positive shock increases the marginal productivity of capital and creates an incentive for higher levels of investment. Some of this effect is mitigated by a counteracting incentive in the penalization of rapid changes in the level of investment in the investment cost adjustment framework. The level of capital stock follows directly from these investment decisions, and similarly, the aggregate consumption and output levels follow as well. This is where the benchmark RBC model and IAC model differ the most: the investment adjustment costs alter the set of incentives faced by a central planner such that the model produces results that more closely resemble the observed stylized facts.

Appendix

Dynare Code

Due to the intricacies of Dynare all state variables must be included with a $t-1$ subscript (Adjemian et al. (2011)). Since current period capital is not a choice variable the production function is rewritten as $Y_t = A_t K_{t-1}^{\alpha} N_t^{1-\alpha}$. This results in the impulse response function associated with capital being lagged a period.

Benchmark RBC Model

```
close all
% Simple RBC Model

var y,c,i,k,n,A;
varexo e;

parameters alpha, beta, delta, rho, psi, sigmae;

alpha    = 0.35;
beta     = 0.97;
delta    = 0.06;
rho      = 0.95;
sigmae   = 0.01;
psi      = 1;

model;

% Standard RBC Model

% Equation 1: Euler Equation for K
1/exp(c)           = beta*(1/exp(c(+1)))*(alpha*exp(A(+1))*exp(k)^(alpha-1)*exp(n(+1))^(1-alpha) + (1-delta));

% Equation 2: Euler Equation for N
psi/(1-exp(n))   = (1/exp(c))   * ((1-alpha)*exp(A)*exp(k)^(alpha)*exp(n)^(-alpha));

% Equation 3: Production Function
exp(y)             = exp(A)*(exp(k(-1))^(alpha))*exp(n)^(1-alpha);
```

```

% Equation 4: Resource Constraint
exp(y)          = exp(c) + exp(k)  - (1-delta)*exp(k(-1));

% Equation 5: Law of Motion for TFP Shocks
A               = rho*A(-1) + e;

% Equation 6: Law of Motion for Capital Accumulation
exp(i)          = exp(k)  - (1-delta)*exp(k(-1));

initval;
c = log(2.5);
y = log(3);
k = log(28);
A = 0 ;
n = log(0.8);
i = log(1.5);
end;

shocks;
var e =  sigmae^2;
end;

steady;

stoch_simul;
```

Investment Adjustment Costs Model

```
close all
% IAC RBC Model

var y,c,i,k,n,A,lambda omega;
varexo e;

parameters alpha, beta, delta, rho, psi, sigmae, phi;

alpha   = 0.35;
beta    = 0.97;
delta   = 0.06;
```

```
rho      = 0.95;
sigmae   = 0.01;
psi      = 1;
phi      = 4.5;

model;

% IAC Model;

% Equation 1: FOC for C
1/exp(c) = exp(lambda);

% Equation 2: FOC for N
psi/(1-exp(n)) = exp(lambda)*(1-alpha)*(exp(y)/exp(n));

% Equation 3: Law of Motion for TFP Shocks
A = rho * A(-1) + e;

% Equation 4: Production Function
exp(y)= exp(A)*(exp(k(-1))^alpha)*exp(n)^(1-alpha);

% Equation 5: Resource Constraint
exp(y) = exp(c) + exp(i);

% Equation 6: Law of Motion for Capital Accumulation (Penalized)
exp(k) = exp(i) - (phi/2)*exp(i)*((exp(i)/exp(i(-1))) -1)^2 + (1-delta)*exp(k(-1)) ;

% Equation 7: FOC for K
exp(omega) = beta*exp(lambda(+1))*(alpha*(exp(y(+1))/exp(k)))+beta*exp(omega(+1))*(1-delta);

% Equation 8: FOC for I
exp(lambda) = exp(omega)-exp(omega)*(phi/2)*((exp(i)/exp(i(-1))-1)^2)-exp(omega)*phi*( (exp(
    i)/exp(i(-1))) -1  )* (exp(i)/exp(i(-1))) + beta*exp(omega(+1))*phi*((exp(i(+1))/exp(i))
    -1)*((exp(i(+1))/exp(i))^2);

end;

initval;
c = log(2.5);
y = log(3);
k = log(28);
```

```
A = 0  ;
n = log(0.8);
i = log(1.5);
omega   = log(0.5);
lambda = log(2);
end;

shocks;
var e =  sigmae^2;
end;

steady;

stoch_simul;
```

Derivations

Benchmark RBC Model

Setting up the problem described by equations (1)-(6) we can derive the following Lagrangian equation:

$$L(C_t, K_{t+1}, N_t, \lambda_t) = E_0 \sum_{t=0}^{\infty} \beta^t \Big\{ \ln C_t + \psi \ln(1 - N_t) + \lambda_t [A_t K_t^{\alpha} N_t^{1-\alpha} - C_t - K_{t+1} + (1-\delta) K_t] \Big\} \tag{13}$$

There are three choice variables $C_t,\ K_{t+1},\ N_t$ and one Lagrange multiplier λ_t. Resultantly, there are three first-order conditions:

$$\frac{\partial L(C_t, K_{t+1}, N_t, \lambda_t)}{\partial C_t} : \frac{1}{C_t} = \lambda_t \tag{14}$$

$$\frac{\partial L(C_t, K_{t+1}, N_t, \lambda_t)}{\partial K_{t+1}} : \lambda_t = \lambda_{t+1} \beta \big[\alpha \frac{Y_{t+1}}{K_{t+1}} + 1 - \delta \big] \tag{15}$$

$$\frac{\partial L(C_t, K_{t+1}, N_t, \lambda_t)}{\partial N_t} : \frac{\psi}{1 - N_t} = \lambda_t (1 - \alpha) \frac{Y_t}{N_t} \tag{16}$$

If we combine equation (14) with equation (15) we obtain the Euler equation for capital. If we combine equation (14) with equation (16) we obtain the Euler equation for labour:

$$\frac{1}{C_t} = \frac{1}{C_{t+1}}\beta\big[\alpha\frac{Y_{t+1}}{K_{t+1}} + 1 - \delta\big] \tag{17}$$

$$\frac{\psi}{1-N_t} = \frac{1}{C_t}(1-\alpha)\frac{Y_t}{N_t} \tag{18}$$

To derive the steady-state solution, we impose that equation (17) is time-invariant and express the capital-output ratio in terms of parameters.

$$\frac{1}{C} = \frac{1}{C}\beta\left[\alpha\frac{Y}{K} + 1 - \delta\right] \implies \frac{Y}{K} = \left[\frac{1}{\beta} - 1 + \delta\right]\frac{1}{\alpha} \tag{19}$$

In steady-state, our law of motion for capital accumulation $K_{t+1} = (1-\delta)K_t + I_t$ will simplify to $K = (1-\delta)K + I$ or simply $\delta K = I$. Combining this with the aggregate resource constraint $C + I = Y$, this will simplify to a consumption-labour ratio of the form:

$$\frac{C}{N} = \frac{Y}{N} - \delta\frac{K}{N} \implies \frac{C}{N} = \left(\frac{K}{N}\right)^{\alpha} - \delta\frac{K}{N} \tag{20}$$

We describe our steady-state production function by $Y = K^{\alpha}N^{1-\alpha}$. Combining this with equation (19), we can obtain:

$$\left[\frac{K}{N}\right]^{\alpha-1} = \left[\frac{1}{\beta} - 1 + \delta\right]\frac{1}{\alpha} \implies \frac{K}{N} = \left(\left[\frac{1}{\beta} - 1 + \delta\right]\frac{1}{\alpha}\right)^{\frac{1}{\alpha-1}} \tag{21}$$

Now we have the steady-state capital-labour ratio in terms of only paramters. We can rewrite equation (18) in a similar fashion:

$$\frac{\psi}{1-N} = \frac{1}{C}(1-\alpha)\frac{Y}{N} \implies \frac{C}{N} = \left(\frac{1-N}{N}\right)\left(\frac{1-\alpha}{\psi}\right)\left(\frac{K}{N}\right)^{\alpha} \tag{22}$$

Since the left-hand side of both equation (20) and equation (22) are the same, we set them equal to observe that, with the fully parameterized capital-labour ratio described by equation (21), can solve for the steady-state level of N fully in terms of parameters:

$$\left(\frac{K}{N}\right)^{\alpha} - \delta\frac{K}{N} = \left(\frac{1-N}{N}\right)\left(\frac{1-\alpha}{\psi}\right)\left(\frac{K}{N}\right)^{\alpha}$$

$$1-\delta\left(\frac{K}{N}\right)^{1-\alpha} = \frac{1-N}{N}\frac{1-\alpha}{\psi}$$

$$N = \frac{1-\alpha}{\psi - \delta\psi\left(\left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{1}{\alpha-1}}\right)^{1-\alpha} + 1-\alpha}$$

$$N^{\mathrm{SS}} = \frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} \tag{23}$$

Thus, N is fully described by parameters and can be substituted into steady-state equations for C, Y, K:

$$K^{\mathrm{SS}} = \left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{1}{\alpha-1}}\frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} \tag{24}$$

$$Y^{\mathrm{SS}} = \left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{\alpha}{\alpha-1}}\frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} \tag{25}$$

$$C^{\mathrm{SS}} = \frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi}\left[\left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{\alpha}{\alpha-1}} - \delta\left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{1}{\alpha-1}}\right] \tag{26}$$

$$I^{\mathrm{SS}} = \delta\left(\left[\frac{1}{\beta}-1+\delta\right]\frac{1}{\alpha}\right)^{\frac{1}{\alpha-1}}\frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} \tag{27}$$

To complete the description of the model, we define several new variables that are functions of other parameters for compactness:

$$\eta = \left[\frac{1}{\beta} - 1 + \delta\right]\frac{1}{\alpha}$$

$$\xi = \frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} \tag{28}$$

Finally, the steady-state values of N, K, Y, C are described fully by parameters:

$$N^{\text{SS}} = \xi \tag{29}$$

$$K^{\text{SS}} = \xi\eta^{\frac{1}{\alpha-1}} \tag{30}$$

$$Y^{\text{SS}} = \xi\eta^{\frac{\alpha}{\alpha-1}} \tag{31}$$

$$C^{\text{SS}} = \xi\left[\eta^{\frac{\alpha}{\alpha-1}} - \eta^{\frac{1}{\alpha-1}}\right] \tag{32}$$

$$I^{\text{SS}} = \delta\xi\eta^{\frac{1}{\alpha-1}} \tag{33}$$

To confirm that these calculations are consistent with the results generated by Dynare, we derive the values of η and ξ by our calibration parameters:

$$\eta = \left[\frac{1}{\beta} - 1 + \delta\right]\frac{1}{\alpha} = 0.2597$$

$$\xi = \frac{(1-\beta+\delta)(1-\alpha)}{(1-\beta+\delta)(\psi+1-\alpha)-\alpha\beta\delta\psi} = 0.4566$$

As can be seen, $N^{\text{SS}} = 0.4566,\ K^{\text{SS}} = 3.644,\ Y^{\text{SS}} = 0.9309,\ C^{\text{SS}} = 0.7123,\ I^{\text{SS}} = 0.2186$ are approximately equivalent to the Dynare results.

Investment Adjustment Costs

Setting up the problem described by equations (7)-(12) we can derive the following Lagrangian equation:

$$L(C_t, K_{t+1}, I_t, N_t, \lambda_t, \omega_t) = E_0 \sum_{t=0}^{\infty} \beta^t \Big\{ \ln C_t + \psi \ln(1 - N_t)$$

$$+\lambda_t[A_t K_t^{\alpha} N_t^{1-\alpha} - C_t - I_t] + \omega_t \Big[[1 - \frac{\phi}{2}[\frac{I_t}{I_{t-1}} - 1]^2] I_t + (1-\delta) K_t - K_{t+1} \Big] \Big\} \tag{34}$$

There are four choice variables $C_t,\ K_{t+1},\ N_t,\ I_t$ and two Lagrange multipliers $\lambda_t,\ \omega_t$. Resultantly, there are four first-order conditions:

$$\frac{\partial L(C_t, K_{t+1}, I_t, N_t, \lambda_t, \omega_t)}{\partial C_t} : \frac{1}{C_t} = \lambda_t \tag{35}$$

$$\frac{\partial L(C_t, K_{t+1}, I_t, N_t, \lambda_t, \omega_t)}{\partial K_{t+1}} : \omega_t = \beta \Big[\lambda_{t+1} [\alpha \frac{Y_{t+1}}{K_{t+1}}] + \omega_{t+1} [1 - \delta] \Big] \tag{36}$$

$$\frac{\partial L(C_t, K_{t+1}, I_t, N_t, \lambda_t, \omega_t)}{\partial N_t} : \frac{\psi}{1 - N_t} = \lambda_t (1-\alpha) \frac{Y_t}{N_t} \tag{37}$$

$$\frac{\partial L(C_t, K_{t+1}, N_t, I_t, \lambda_t, \omega_t)}{\partial I_t} : \lambda_t = \omega_t \Big[1 - \frac{\phi}{2} \Big(\frac{I_t}{I_{t-1}} - 1 \Big)^2 - \phi \frac{I_t}{I_{t-1}} [\frac{I_t}{I_{t-1}} - 1] \Big]$$

$$+ \beta \omega_{t+1} \Big[\phi (\frac{I_{t+1}}{I_t})^2 (\frac{I_{t+1}}{I_t} - 1) \Big] \tag{38}$$

To solve the model in steady-state, we combine equations and impose that all variables are time-invariant. The easiest way to proceed is to consider equation (38):

$$\lambda = \omega \Big[1 - \frac{\phi}{2} \Big(\frac{I}{I} - 1 \Big)^2 - \phi \frac{I}{I} [\frac{I}{I} - 1] \Big] + \beta \omega \Big[\phi (\frac{I}{I})^2 (\frac{I}{I} - 1) \Big] \implies \lambda = \omega$$

By the result that $\omega = \lambda$, it is easy to see that our steady-state production function and capital-output ratio will be equivalent to those given in the benchmark model. Proving that the penalized law of motion for capital accumulation simplifies to $\delta K = I$ is sufficient in proving that the steady-state values will be equivalent in the IAC and RBC frameworks:

$$K = (1-\delta)K + \big[1 - \frac{\phi}{2}[\frac{I}{I} - 1]^2\big]I \implies K = (1-\delta)K + I \implies I = \delta K$$

Resultantly, all of our steady-state equations are identical in the IAC model to the benchmark RBC model. Thus, the steady state equations (29)-(33) hold and follow the exact same method of solution for both models.

References

Adjemian, Stéphane, Houtan Bastani, Michel Juillard, Ferhat Mihoubi, George Perendia, Marco Ratto, and Sébastien Villemot. 2011. "Dynare: Reference Manual, Version 4." Citeseer.

Christiano, Lawrence J., Martin Eichenbaum, and Charles L. Evans. 2005. "Nominal Rigidities and the Dynamic Effects of a Shock to Monetary Policy." *Journal of Political Economy* 113 (1): 1–45. https://ideas.repec.org/a/ucp/jpolec/v113y2005i1p1-45.html.

Cogley, Timothy, and James M Nason. 1995. "Output Dynamics in Real-Business-Cycle Models." *American Economic Review* 85 (3): 492–511. https://ideas.repec.org/a/aea/aecrev/v85y1995i3p492-511.html.

Doms, Mark, and Timothy Dunne. 1998. "Capital Adjustment Patterns in Manufacturing Plants." *Review of Economic Dynamics* 1 (2): 409–29. http://EconPapers.repec.org/RePEc:red:issued:v:1:y:1998:i:2:p:409-429.

Fiori, Giuseppe. 2012. "Lumpiness, Capital Adjustment Costs and Investment Dynamics." *Journal of Monetary Economics* 59 (4). Elsevier: 381–92.

King, Robert G., and Sergio T. Rebelo. 2000. "Resuscitating Real Business Cycles." RCER Working Papers 467. University of Rochester - Center for Economic Research (RCER). https://ideas.repec.org/p/roc/rocher/467.html.

Smets, Frank, and Raf Wouters. 2003. "An Estimated Dynamic Stochastic General Equilibrium Model of the Euro Area." *Journal of the European Economic Association* 1 (5): 1123–75. http://EconPapers.repec.org/RePEc:tpr:jeurec:v:1:y:2003:i:5:p:1123-1175.

Torres, Jose. 2015. *Introduction to Dynamic Macroeconomic General Equilibrium Models.* 2nd ed. Vernon

Art; Science Inc. http://EconPapers.repec.org/RePEc:vpr:ecbook:54.

Uribe, Martín, and Stephanie Schmitt-Grohé. 2005. “Optimal Fiscal and Monetary Policy in a Medium Scale Macro Model.” Computing in Economics and Finance 2005 476. Society for Computational Economics. http://EconPapers.repec.org/RePEc:sce:scecf5:476.